Creating a Landscape Report with ODS PDF Generate a PDF document

in landscape orientation. Print a report and map side by side.

1. Open **p106p03.sas** from the **practices** folder. Run the program and examine the output. The program produces a table and a map for North Atlantic region storms in the 2016 season. **Note**: You must have SAS 9.4M5 to run this code.

2. Modify the program to produce a PDF file named **StormSummary.pdf** in the output folder in the course files. Set the output style to **Journal**. Specify **not** to generate bookmarks in the file.

```
ods pdf file="&outpath/StormSummary.PDF" style=Journal nobookmarkgen;

title1 "2016 Northern Atlantic Storms";

proc sgmap plotdata=pg1.storm_final;
    *openstreetmap;
    esrimap url='http://services.arcgisonline.com/arcgis/rest/services/World_Physical_Map';
    bubble x=lon y=lat size=maxwindmph / datalabel=name datalabelattrs=(color=red size=8);
    where Basin='NA' and Season=2016;
    keylegend 'wind';
run;

proc print data=pg1.storm_final noobs;
    var name StartDate MaxWindMPH StormLength;
    where Basin="NA" and Season=2016;
    format StartDate monyy7.;
run;

ods pdf close;
```

3. Use SAS Help to find a SAS system option that changes the page layout to landscape.

```
options orientation=landscape;
ods pdf file="&outpath/StormSummary.PDF" style=Journal nobookmarkgen;

title1 "2016 Northern Atlantic Storms";

proc sgmap plotdata=pg1.storm_final;
    *openstreetmap;
    esrimap url='http://services.arcgisonline.com/arcgis/rest/services/World_Physical_Map';
    bubble x=lon y=lat size=maxwindmph / datalabel=name datalabelattrs=(color=red size=8);
    where Basin='NA' and Season=2016;
    keylegend 'wind';
run;

proc print data=pg1.storm_final noobs;
    var name StartDate MaxWindMPH StormLength;
    where Basin="NA" and Season=2016;
    format StartDate monyy7.;
run;

ods pdf close;
```

4. Use SAS Help to learn about the ODS LAYOUT GRIDDED statement as a way that you can control the layout of multiple result objects. Force the results to be arranged in one row and two columns.

```
options orientation=landscape;
ods pdf file="&outpath/StormSummary.PDF" style=Journal nobookmarkgen;

title1 "2016 Northern Atlantic Storms";

ods layout gridded columns=2 rows=1;
ods region;

proc sgmap plotdata=pg1.storm_final;
    *openstreetmap;
```

```
    esrimap url='http://services.arcgisonline.com/arcgis/rest/services/World_Physical_Map';
    bubble x=lon y=lat size=maxwindmph / datalabel=name datalabelattrs=(color=red size=8);
    where Basin='NA' and Season=2016;
    keylegend 'wind';
run;

ods region;
proc print data=pg1.storm_final noobs;
    var name StartDate MaxWindMPH StormLength;
    where Basin="NA" and Season=2016;
    format StartDate monyy7.;
run;

ods layout end;
ods pdf close;
```

5. Reset the system option at the end of the program so that future results have a portrait layout.

```
options orientation=landscape;
ods pdf file="&outpath/StormSummary.PDF" style=Journal nobookmarkgen;

title1 "2016 Northern Atlantic Storms";

ods layout gridded columns=2 rows=1;
ods region;

proc sgmap plotdata=pg1.storm_final;
    *openstreetmap;
    esrimap url='http://services.arcgisonline.com/arcgis/rest/services/World_Physical_Map';
    bubble x=lon y=lat size=maxwindmph / datalabel=name datalabelattrs=(color=red size=8);
    where Basin='NA' and Season=2016;
    keylegend 'wind';
run;

ods region;
proc print data=pg1.storm_final noobs;
    var name StartDate MaxWindMPH StormLength;
    where Basin="NA" and Season=2016;
    format StartDate monyy7.;
run;

ods layout end;
ods pdf close;
options orientation=portrait;
```