

Exporting Results

1. Create the PROC EXPORT step to read the **pg1.storm_final** SAS table and create a comma-delimited file named **storm_final.csv**. Use **&outpath** to provide the path to the file.

```
proc export data=pg1.storm_final  
  outfile="&outpath/storm_final.csv"  
  dbms=csv replace;  
run;
```

2. Create the LIBNAME statement using the XLSX engine to create an Excel workbook named **storm.xlsx**. Write the **storm_final** table to the **storm.xlsx** file. Use **&outpath** to provide the path to the file.

```
libname xl_lib xlsx "&outpath/storm.xlsx";  
  
data xl_lib.storm_final;  
  set pg1.storm_final;  
  drop Lat Lon Basin OceanCode;  
run;  
  
libname xl_lib clear;
```

3. Add ODS statements to create an Excel file named **pressure.xlsx**. Use **&outpath** to provide the path to the file. Be sure to close the ODS location at the end of the program. Also add the **STYLE=ANALYSIS** option in the first ODS EXCEL statement.

```
ods excel file="&outpath/pressure.xlsx" style=analysis;  
  
title "Minimum Pressure Statistics by Basin";  
ods noproctitle;  
proc means data=pg1.storm_final mean median min maxdec=0;  
  class BasinName;  
  var MinPressure;  
run;  
  
title "Correlation of Minimum Pressure and Maximum Wind";  
proc sgscatter data=pg1.storm_final;  
  plot minpressure*maxwindmph;  
run;  
title;  
  
ods proctitle;  
ods excel close;
```