

Creating an Excel File Using ODS EXCEL

Create an Excel workbook named **StormStats.xlsx** that includes the results of SAS procedures. Customize the names of the Excel worksheets.

1. Open **p106p01.sas** from the **practices** folder. Before the PROC MEANS step, add an ODS EXCEL statement to do the following:
 - Write the output file to **"&outpath/StormStats.xlsx"**.
 - Set the style for the Excel file to **snow**.
 - Set the sheet name for the first tab to **South Pacific Summary**.

```
ods excel file="&outpath/StormStats.xlsx"
  style=snow
  options(sheet_name='South Pacific Summary');

proc means data=pg1.storm_detail maxdec=0 median max;
  class Season;
  var Wind;
  where Basin='SP' and Season in (2014,2015,2016);
run;

proc print data=pg1.storm_detail noobs;
  where Basin='SP' and Season in (2014,2015,2016);
  by Season;
run;
```

2. Turn off the procedure titles and report titles at the start of the program. Turn the procedure titles on at the end of the program.

```
ods excel file="&outpath/StormStats.xlsx"
  style=snow
  options(sheet_name='South Pacific Summary');
ods noproctitle;

proc means data=pg1.storm_detail maxdec=0 median max;
  class Season;
  var Wind;
  where Basin='SP' and Season in (2014,2015,2016);
run;

proc print data=pg1.storm_detail noobs;
  where Basin='SP' and Season in (2014,2015,2016);
  by Season;
run;

ods proctitle;
```

3. Immediately before the PROC PRINT step, add an ODS EXCEL statement to change the sheet name to **Detail**.

```
ods excel file="&outpath/StormStats.xlsx"
  style=snow
  options(sheet_name='South Pacific Summary');
ods noproctitle;

proc means data=pg1.storm_detail maxdec=0 median max;
  class Season;
  var Wind;
  where Basin='SP' and Season in (2014,2015,2016);
run;

ods excel options(sheet_name='Detail');

proc print data=pg1.storm_detail noobs;
  where Basin='SP' and Season in (2014,2015,2016);
  by Season;
run;

ods proctitle;
```

4. At the end of the program, add an ODS EXCEL statement to close the Excel destination.

```
ods excel file="&outpath/StormStats.xlsx"
  style=snow
  options(sheet_name='South Pacific Summary');
ods noproctitle;

proc means data=pg1.storm_detail maxdec=0 median max;
  class Season;
  var Wind;
  where Basin='SP' and Season in (2014,2015,2016);
run;

ods excel options(sheet_name='Detail');

proc print data=pg1.storm_detail noobs;
  where Basin='SP' and Season in (2014,2015,2016);
  by Season;
run;

ods excel close;
ods proctitle;
```